

Casio fx-7700GE
Casio fx-9700GE
Casio CFX-9800G
Casio CFX-9850G

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists. Solutions to systems of linear equations are also available directly from the Casio calculator's EQUATION MENU.

SOLVE

"AX+BY=C" ↓

"A=": ? → A ↓

"B=": ? → B ↓

"C=": ? → C ↓

"DX+EY=F" ↓

"D=": ? → D ↓

"E=": ? → E ↓

"F=": ? → F ↓

AE-DB=0 ⇒ Goto 1 ↓

"X=": (CE-BF) ÷ (AE-DB) ▲

"Y=": (AF-CD) ÷ (AE-DB) ↓

Goto 2 ↓

Lbl 1 ↓

"NO UNIQUE SOLUTION" ↓

Lbl 2

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y -intercepts of the lines are between -10 and 10 .)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension Mat A as a 2×3 matrix. Press EXE after each screen display to continue the program.

```

ROWOPS
"ENTER A"↓
"2 BY 3 MATRIX:"↓
"A B C"↓
"D E F"↓
"A="?"→A:"B="?"→B:
"C="?"→C:"D="?"→D:
"E="?"→E:"F="?"→F:↓
[[A,B,C][D,E,F]]→Mat A↓
Cls↓
"ORIGINAL MATRIX:"▲
Mat A▲
Range -10,10,1,-10,10,1↓
Graph Y=B-1(C-AX)↓
Graph Y=E-1(F-DX)▲
Cls↓
"OBTAIN LEADING"↓
"1 IN ROW 1"▲
*Row A-1,A,1↓
Mat A▲
Graph Y=(A÷B)(C÷A-X)↓
Graph Y=E-1(F-DX)▲
Cls↓
"OBTAIN 0 BELOW"↓
"LEADING 1 IN"↓
"COLUMN 1"▲
*Row+ -D,A,1,2↓
Mat A▲
Graph Y=(A÷B)(C÷A-X)↓
Graph Y=(E-(BD÷A))-1(F-(DC÷A))▲
Cls↓
Mat A[2,2]→G↓
G=0⇒Goto 1↓
*Row G-1,A,2↓
"OBTAIN LEADING"↓
"1 IN ROW 2"▲
Mat A▲
Graph Y=(A÷B)(C÷A-X)↓
Graph Y=(E-(BD÷A))-1(F-(DC÷A))▲
Cls↓
"OBTAIN 0 ABOVE"↓
"LEADING 1 IN"↓
"COLUMN 2"▲
Mat A[1,2]→H↓
*Row+ -H,A,2,1↓
Mat A▲
Mat A[1,3]→J↓
Mat A[2,3]→K↓
Graph Y=K↓
Plot J,-10:Plot J,10:Line▲
"THE POINT OF"↓
"INTERSECTION IS"↓
"X=":J▲
"Y=":K
Goto 3↓
Lbl 1↓
Mat A[2,3]=0⇒Goto 2↓
"INCONSISTENT"↓
"SYSTEM"↓
Goto 3↓
Lbl 2↓
"INFINITELY MANY"↓
"SOLUTIONS"↓
Lbl 3

```